<!DOCTYPE HTML>
<html lang="en">
<head>
  <meta name="copyright" content=
  "Copyright (c) IBM Corporation and others 2000, 2009. This page is made available under license. For full details see the LEGAL in the documentation book that contains this page.">
  <meta charset="utf-8">
  <link rel="STYLESHEET" href="../book.css" type="text/css">
  <title>antRunner application entry point</title>
</head>
<body>
  <h1>Antrunner Application Entry Point</h1>
  <p>The Ant core plug-in declares a main entry point that allows the running of Eclipse's integrated Ant support in a
  headless environment. An Ant build can then be run directly from the main platform launcher.</p>
  <p>This is an example <code>.bat</code> file that shows how to make use of the antRunner application entry point on
  Windows.</p>
  <pre> 
echo off
setlocal

REM ****************************************************

REM The JRE java.exe to be used
set JAVAEXE="C:\jdk1.4.2\jre\bin\java.exe"

REM The Eclipse launcher jar
set LAUNCHERJAR="C:\eclipse\plugins\org.eclipse.equinox.launcher_1.0.0.v20070319.jar"

REM The location of your workspace
set WORKSPACE=C:\runtime-workspace

REM The buildfile to use for the build
set BUILDFILE=build.xml

REM ****************************************************

if not exist %JAVAEXE% echo ERROR: incorrect java.exe=%JAVAEXE%, edit this file and correct the JAVAEXE envar
if not exist %JAVAEXE% goto done

if not exist %LAUNCHERJAR% echo ERROR: incorrect equinox.launcher jar=%LAUNCHERJAR%, edit this file and correct the LAUNCHERJAR envar
if not exist %LAUNCHERJAR% goto done

if not exist %WORKSPACE% echo ERROR: incorrect workspace=%WORKSPACE%, edit this file and correct the WORKSPACE envar
if not exist %WORKSPACE% goto done

if not exist %BUILDFILE% echo ERROR: incorrect buildfile=%BUILDFILE%, edit this file and correct the BUILDFILE envar
if not exist %BUILDFILE% goto done

:run
@echo on
%JAVAEXE% -cp %LAUNCHERJAR% org.eclipse.equinox.launcher.Main -noupdate -application org.eclipse.ant.core.antRunner -data %WORKSPACE% -buildfile %BUILDFILE%

:done
pause
</pre>
  <h3 class="related">Related Concepts</h3><a href="../concepts/concepts-antsupport.htm">Ant support</a>
</body>
</html>
